
	  function fl_SDP_SelectedEVSCodecParameters_Common(SDP_media_desc p_MediaDescription_InitialOffer) return SemicolonParam_List
  {
    var SemicolonParam_List v_FmtpParameterList_InitialOffer := f_SDP_MediaDescr_GetFirstEVSCodecParameters(p_MediaDescription_InitialOffer);
    var template (value) SemicolonParam_List v_FmtpParameterList;
    var charstring v_BrVal;
    var charstring v_BwVal;

    if (match(v_FmtpParameterList_InitialOffer, cr_EVS_Fmtp_AudioParamsBrBw("13.2", "swb"))) {
      v_BrVal := "13.2";                  // 34.229-5 A.4.1/2 183 Session Progress, NOTE 8
      v_BwVal := "swb";
    } else {
      v_BrVal := "5.9-13.2";              // 34.229-5 A.4.1/2 183 Session Progress, NOTE 9
      v_BwVal := "nr-swb";
    }
    v_FmtpParameterList := { cs_EVS_Br(v_BwVal), cs_EVS_Bw("swb"), cs_Fmtp_ModesetParam("0,1,2") };
    return valueof(v_FmtpParameterList);
  }

	  function f_SDP_SelectedEVSCodecParameters_AnswerTX(SDP_media_desc p_MediaDescription_InitialOffer) return template (value) SemicolonParam_List
  {
    var template (value) SemicolonParam_List v_FmtpParameterList;

    v_FmtpParameterList := fl_SDP_SelectedEVSCodecParameters_Common(p_MediaDescription_InitialOffer);
    v_FmtpParameterList := f_SemicolonParam_List_Add_TX(v_FmtpParameterList, cs_AMR_MaxRed);
    
    return v_FmtpParameterList;
  }

	  function f_IMS_SDP_AudioMediaAttributes_InitialAnswer_TX(SDP_media_desc p_MediaDescription_Offer,
                                                           IMS_SDP_AudioCodec_Type p_AudioCodec,
                                                           charstring p_FmtAudio)
    runs on IMS_PTC
    return SDP_attribute_list
  {
    var template (omit) SDP_attribute_list v_SDP_MediaAttributes;
    var template (omit) SDP_attribute v_Attribute;
    var template (omit) SemicolonParam_List v_FmtpParameterList_Answer := omit;
    var SemicolonParam_List v_FmtpParameterList_Offer;
    var CharStringList_Type v_FmtpParametersToBoCopied;
    var charstring v_BwVal;

    select (p_AudioCodec) {                     // @sic R5-151968, R5-153999 sic@
      case (EVS_16000){
        if (f_IMS_PTC_RanTypeIsNR5GC()) {       // @sic R5-206382: implementation of TC_7_4 sic@
          v_FmtpParameterList_Answer := f_SDP_SelectedEVSCodecParameters_AnswerTX(p_MediaDescription_Offer);
          v_FmtpParameterList_Offer := f_SDP_MediaDescr_GetFirstEVSCodecParameters(p_MediaDescription_Offer);
          if (match(v_FmtpParameterList_Offer, cr_EVS_Fmtp_AudioParamsBrBw("13.2", "swb"))) {
            v_BwVal := "13.2";                  // 34.229-5 A.4.1/2 183 Session Progress, NOTE 8
          } else {
            v_BwVal := "5.9-13.2";              // 34.229-5 A.4.1/2 183 Session Progress, NOTE 9
          }
          v_FmtpParameterList_Answer := { cs_EVS_Br(v_BwVal), cs_EVS_Bw("swb"), cs_Fmtp_ModesetParam("0,1,2"), cs_AMR_MaxRed };
        }
        else {                                  // C.44 @sic R5-153999 sic@
          v_Attribute := f_SDP_AttributeList_GetAttribute(p_MediaDescription_Offer.attributes, cr_SDP_Attribute_fmtp(p_FmtAudio));
          v_BwVal := f_SIP_SemicolonParamList_GetParamValue(v_Attribute.fmtp.params.paramList, "bw");   /* @sic R5s170336/R5-172069 sic@ */
          v_FmtpParameterList_Answer := {cs_EVS_Bw(v_BwVal), cs_EVS_BwSend(v_BwVal), cs_EVS_BwRecv(v_BwVal), cs_AMR_MaxRed};   /* @sic R5s160379: Additional Change by MCC160 sic@
                                                                                                                                  @sic R5s170336/R5-172069: v_BwVal sic@ */
          v_FmtpParametersToBoCopied := { "br", "br-send", "br-recv" };
          v_FmtpParameterList_Offer := f_SDP_MediaDescr_GetCodecParameters(p_MediaDescription_Offer.attributes, p_FmtAudio);
          v_FmtpParameterList_Answer := f_SIP_SemicolonParamList_CopyParams(v_FmtpParameterList_Offer, v_FmtpParameterList_Answer, v_FmtpParametersToBoCopied);
        }
        v_SDP_MediaAttributes := f_SDP_MediaAttributes_EVS_AudioDef(p_FmtAudio, valueof(v_FmtpParameterList_Answer));   // valueof to avoid compiler warning
      }
      case (AMR_16000){                         // @sic R5-153950 sic@
        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AMR_WB_AudioDef(p_FmtAudio);  /* @sic R5s160379 change 2 sic@ */
      }
      case (AMR_8000){
        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AudioDef(p_FmtAudio);   /* @sic R5-144693: new function f_SDP_MediaAttributes_AudioDef sic@ */
      }
    }
    return valueof(v_SDP_MediaAttributes);
  }





